
HP-35s Calculator Program Curves 2A
Compute Values for a Parabolic Vertical Curve

Programmer: Dr. Bill Hazelton

Date: March, 2008. Version: 1.0 Mnemonic:  P for Parabolic Vertical Curve.

Line Instruction Display User Instructions
P001 LBL P   LBL  P
P002 CLSTK   CLEAR  5
P003 FS?  10   FLAGS  3  .0
P004 GTO  P008
P005 SF  1   FLAGS  1  1
P006 SF  10   FLAGS  1  .0
P007 GTO  P009
P008 CF  1   FLAGS  2  1
P009 COMP  V  CURVE (Key in using EQN RCL C, RCL O,  etc.)
P010 PSE   PSE
P011 CLx   CLEAR  1
P012 STO  S   STO  S
P013 STO  R   STO  R
P014 STO  Q   STO  Q
P015 STO  P   STO  P
P016 STO  A   STO  A
P017 STO  L   STO  L
P018 ENTER  P  I  RD (Key in using EQN RCL E, RCL N,  etc.)
P019 PSE   PSE
P020 INPUT  R R?   INPUT  R
P021 ENTER  P  I  EL (Key in using EQN RCL E, RCL N,  etc.)
P022 PSE   PSE
P023 INPUT  S S?   INPUT  S
P024 START GRADE (Key in using EQN RCL S, RCL T,  etc.)
P025 PSE   PSE
P026 INPUT  P P?   INPUT  P
P027 END GRADE (Key in using EQN RCL E, RCL N,  etc.)
P028 PSE   PSE
P029 INPUT  Q Q?   INPUT  Q
P030 RCL  Q
P031 RCL—  P
P032 STO  A   STO  A
P033 ENTER  LENGTH (Key in using EQN RCL E, RCL N,  etc.)
P034 PSE   PSE
P035 INPUT  L L?   INPUT  L
P036 RCL  R
P037 RCL  L
P038 2
P039 ÷
P040 —

Line Instruction Line Instruction Line Instruction
P041 STO  U P082 RCL+  I P122 RCL  Y
P042 RCL  S P083 STO  D P123 x ≤ 0?
P043 RCL  P P084 VIEW  D P124 GTO  P149
P044 RCL×  L **** End Pts Data P125 ENTER RD
P045 2 P085 0 P126 PSE
P046 ÷ P086 STO  Y P127 INPUT  X
P047 — P087 END  PTS  [0–1] P128 RCL  X
P048 STO  V P088 PSE P129 RCL—  U
P049 RCL  A P089 INPUT  Y P130 STO  I
P050 RCL÷  L P090 RCL  Y P131 x2

P051 2 P091 x ≤ 0? P132 RCL×  B
P052 ÷ P092 GTO  P116 P133 RCL  I
P053 STO  B P093 START PT RD P134 RCL×  P
P054 RCL  P P094 PSE P135 +
P055 RCL÷  B P095 VIEW  U P136 RCL+  V
P056 2 P096 START  PT  EL P137 STO  H
P057 ÷ P097 PSE P138 EL  OF  POINT
P058 +/— P098 VIEW  V P139 PSE
P059 STO  I P099 END  PT  RD P140 VIEW  H
**** Max/Min Pt. Data P100 PSE P141 0
P060 0 P101 RCL  U P142 STO  Y
P061 STO  Y P102 RCL+  L P143 AGAIN  [0–1]
P062 MAX–MIN [0–1] P103 STO  F P144 PSE
P063 PSE P104 VIEW  F P145 INPUT  Y
P064 INPUT  Y P105 RCL  L P146 RCL  Y
P065 RCL  Y P106 x2 P147 x > 0?
P066 x ≤ 0? P107 RCL×  B P148 GTO  P125
P067 GTO  P085 P108 RCL  L **** Get RD from EL
P068 RCL  I P109 RCL×  P P149 0
P069 x2 P110 + P150 STO  Y
P070 RCL×  B P111 RCL+  V P151 STO  H
P071 RCL  I P112 STO  E P152 STO  X
P072 RCL×  P P113 END  PT  EL P153 COMP  1  RD [0–1]
P073 + P114 PSE P154 PSE
P074 RCL+  V P115 VIEW  E P155 INPUT  Y
P075 STO  E **** Get EL from RD P156 RCL  Y
P076 MAX—MIN  EL P116 0 P157 x ≤ 0?
P077 PSE P117 STO  Y P158 GTO  P224
P078 VIEW  E P118 STO  X P159 ENTER  EL
P079 MAX—MIN RD P119 COMP  1  EL [0–1] P160 PSE
P080 PSE P120 PSE P161 INPUT  H
P081 RCL  U P121 INPUT  Y P162 RCL  V

****  These lines are simply comments in the code. You don’t key it into the calculator!

Line Instruction Line Instruction Line Instruction
P163 RCL—  H P207 STO  X P250 VIEW  V
P164 STO  C P208 SINGLE  RD P251 RD VALUE
P165 RCL  P P209 PSE P252 PSE
P166 x2 P210 VIEW  X P253 RCL  C
P167 RCL  B P211 GTO  P216 P254 RCL+  U
P168 RCL×  C P212 THIS ELEV P255 STO  X
P169 4 P213 PSE P256 VIEW  X
P170 × P214 NOT ON CURVE P257 RCL  C
P171 — P215 PSE P258 x2

P172 STO  D P216 0 P259 RCL×  B
P173 x < 0? P217 STO  Y P260 RCL  C
P174 GTO  P212 P218 AGAIN  [0–1] P261 RCL×  P
P175 x = 0? P219 PSE P262 +
P176 GTO  P201 P220 INPUT  Y P263 RCL+  V
P177 RCL  D P221 RCL  Y P264 STO  H
P178 √x P222 x > 0? P265 EL  VALUE
P179 STO  D P223 GTO  P159 P266 PSE
P180 RCL—  P **** Step thru RDs P267 VIEW  H
P181 RCL÷  B P224 0 P268 RCL  C
P182 2 P225 STO  Y P269 STO  I
P183 ÷ P226 STO  X P270 RCL  D
P184 RCL+  U P227 STO  H P271 STO+  I
P185 STO  X P228 STO  D P272 RCL  I
P186 FIRST  RD P229 STO  C P273 RCL—  L
P187 PSE P230 STO  I P274 x > 0?
P188 VIEW  X P231 STEP THRU RD P275 GTO  P294
P189 RCL  D P232 PSE P276 RCL  I
P190 +/— P233 NO—YES  [0–1] P277 x2

P191 RCL—  P P234 PSE P278 RCL×  B
P192 RCL÷  B P235 INPUT  Y P279 RCL  I
P193 2 P236 RCL  Y P280 RCL×  P
P194 ÷ P237 x ≤ 0? P281 +
P195 RCL+  U P238 GTO  P313 P282 RCL+  V
P196 STO  X P239 FIRST INCRMNT P283 STO  H
P197 SECOND  RD P240 PSE P284 RCL  I
P198 PSE P241 INPUT  C P285 RCL+  U
P199 VIEW  X P242 GENRL INCRMNT P286 STO  X
P200 GTO  P216 P243 PSE P287 RD VALUE
P201 RCL  P P244 INPUT  D P288 PSE
P202 +/— P245 RD  VALUE P289 VIEW  X
P203 RCL÷  B P246 PSE P290 EL  VALUE
P204 2 P247 VIEW  U P291 PSE
P205 ÷ P248 EL  VALUE P292 VIEW  H
P206 RCL+  U P249 PSE P293 GTO  P270

Line Instruction
P294 END POINT
P295 PSE
P296 RCL  L
P297 x2

P298 RCL×  B
P299 RCL  L
P300 RCL×  P
P301 +
P302 RCL+  V
P303 STO  H
P304 RCL  L
P305 RCL+  U
P306 STO  X
P307 RD  VALUE
P308 PSE
P309 VIEW  X
P310 EL  VALUE
P311 PSE
P312 VIEW  H
**** End part
P313 PROGRAM  END
P314 PSE
P315 FS?  1
P316 CF  10
P317 STOP
P318 RTN

Notes

1. The **** lines are comments and are not to be entered into the calculator. They are there
to make it easier to work through entering a long program.

2. The program offers options for the processing to be done at each step, asking the user if a
particular set of operations are to be done. The user is prompted for a yes/no answer, with
the input variable being Y, which is set to 0 (no) by default. To skip the operation, just
press R/S. To do the operation, key in 1, then press R/S. The sense is that 0 = no, 1 = yes.
Note that there is a need to pause after the prompt, otherwise the entered value will not
make it into the calculator, and the ‘no’ response will be acted upon.

3. Because grades are used, it is important that the units for elevation and running distance
are the same. Otherwise the computations will not be correct.

4. Some of the calculations will yield results that are on the parabola being used, but not
within the actual segment being used. This is particularly the case with finding maximum
and minimum points, which may not lie within the limits of the curve, and finding running
distances given elevations. The user should check that the running distances fall within the

start and end points of the curve and ignore results that lie outside the curve. The program
does not check these limits.

Theory and Background

The theory of computing the various values for a parabolic ‘equal-tangent-length’ vertical curve
is fairly straightforward. There are six basic possibilities of how the vertical curve can be placed
between two grade lines, as shown in the figure below. The curve allows the grade to be changed
smoothly from the incoming grade (p) to the outgoing grade (q).

+p%

–p% +q%

–q%

Difference in gradients
( q – p )

Difference in gradients
( q – p )

+p%

+q%

–p%

–q%

Difference in gradients
( q – p )

–p%

–q%

+p%

+q%Difference in gradients
( q – p )

The convention with vertical curves is that they are drawn and computed going from left to right.
If the grade is rising from left to right, the gradient is positive. If it is going down from left to
right, the gradient is negative. The above diagram shows the signs that the various gradients
would take in the circumstances.

If the grades on opposite sides of the curve have the same sign, as in the two examples in the right
in the figure above, there is no maximum or minimum value along the curve (other than at its end
points). If the grades have opposite signs (as in the two examples on the left side of the above
diagram), then there will be a maximum or minimum point of the parabola somewhere along the

curve. In this case, the calculation of the elevation and location of such a point is meaningful. If
such a calculation is made for the parabolas on the right in the diagram, the turning point for the
parabola will be determined, but it will be well outside the part of the parabolic curve actually
used. For example, the parabolas shown on the right side of the diagram above have their turning
points well to the right of the end of the curve.

For the purposes of vertical curve design, parabolas are the preferred curve. They are simpler to
compute than circles or ellipses, but differ from them by amounts that are too small to matter in
almost all cases. With a parabola, there is a constant change of grade (or gradient) around the
curve, whereas for a circle there is a constant change of angle around the curve.

Another useful characteristic of the parabola is that it can be placed so that the lengths of the
tangents are always equal. As the tangents are usually close to level (grades usually being fairly
small), the horizontal distances from the point of intersection of the two grade lines to the tangent
points are equal, and half the length of the curve. This allows easy placement of the curve with
respect to the point of intersection of the grade lines. [Note that if the in and out grades are not
equal and of opposite sign, the maximum or minimum point will not fall directly below the
intersection point.]

Gradients, Grades, Slopes and Angles

The slopes of the lines into and out of the vertical curve may be expressed in several ways. For
this program, gradients should be entered as a decimal value of the gradient.

The gradient is the value of the change in elevation over a horizontal distance divided by that
distance, i.e., rise over run, expressed as a decimal number. So, if the slope rises by 2 units for
every 100 along, the gradient is +2 ÷ 100  =  +0.02. If the slope falls by 5 units for 125 units
along, the gradient is  –5 ÷ 125  =  –0.04.

The gradient can also be expressed as a percentage, which is simply the gradient value (as above)
multiplied by 100 to convert it to a percentage. So the above examples would be +2% and –4%,
respectively. Percentages are also handy in that they link in well to horizontal distances expressed
in stations. As the distance is then in 100 ft units, a 1 foot rise would be a +1% grade, so the rise
or fall over one 100 ft ‘station’ can be converted directly to the percentage gradient.

Gradients can also be expressed as a ratio of the rise to the run, and expressed in the form “1 in so
many.” To get this “so many” value, simply calculate the reciprocal value of the gradient (as a
decimal), so for the two example given above, +0.02 would be 1⁄0.02  =  50, and so +1 in 50; –0.04
would be 1⁄0.04  =  25, and so –1 in 25.

Gradients can also be represented as the angle of the line from the horizontal, usually given in
decimal degrees. The tangent of this angle will be the gradient, and the gradient can be converted
to an angle by taking the arctangent of the gradient. So a gradient of +0.02 will give an angle of
arctan(+0.02) = +1.146°, while –0.04 will give and angle of arctan(–0.04) = –2.291°. A slope of
1° would give a gradient of tan(1°) = 0.017.

Grade and gradient are used interchangeably, although sometimes they are applied to specific
representations. For this program, convert all gradients to the decimal format, e.g., +0.02, –0.04.
Be aware that the sign of the gradient is very important and must be included.

The term ‘slope’ is also used, but it usually doesn’t refer to a specific representation.

Horizontal Distances

Horizontal distances as used in the construction of linear objects are commonly expressed as a
distance from a starting point somewhere along the object. How they are expressed depends upon
the units being used, the country in which they are being used, and local practice. Similarly, what
they are called also varies.

Distances in feet are commonly recorded as ‘station’ values. Here, it is assumed that a station is
marked every 100 ft, and that the stations are numbered sequentially from the start, with distances
on from the station noted as additional distance. So a distance of 12,546.78 ft would be recorded
as 125 + 46.78, meaning 125 stations of 100 feet, plus 46.78 feet past that station. For many
construction projects, having points every 100 ft (30.48 m) is very convenient, hence the
popularity of this representation. It is easy to convert between the distance representation
(12,546.78 ft) and the station representation (125 + 46.78): simply remove the + sign and place
the digits together to go to distance, or open the digits two left of the decimal point and put in a +,
to convert to stations. (Calculators prefer the distance version.)

With metric units, 100 m is a bit long for station placement, so an equivalent metric
representation never really caught on. In metric, it is more usual to use the distance representation
(in meters) for all uses. This is also simpler as surveying moves away from the reliance on short
lines for set-out (greatly aided by a station every 100 ft), to total station based set-out by
co-ordinates across large areas.

The distances are known as ‘stations’ (when working with the 100 ft units), but this is a little odd
when using a distance representation. In this case, the distance may be known as the ‘Running
Distance’ (abbreviated RD), or the ‘chainage’ in some circumstances. As the station
representation is so easily converted to the distance representation, and this program can also be
used for metric applications, the horizontal distances in this program will be termed ‘Running
Distances’ and often noted as RD.

Calculating The Parabola

The general equation of a parabola is:

y  =  ax2  +  bx  +  c

The magnitude of the term a controls the sharpness of the parabola, while the sign of a controls
the orientation. With a positive, the parabola is turned upwards and is bowl-shaped (the apex or
turning point is the smallest y value), while with a negative, the parabola is hill-shaped, with the
apex having the largest y value. So a summit or crest has a negative a, while a sag has a positive
a.

It is convenient to use the starting point of the parabola (i.e., the first or left-most tangent point)
as the origin of the co-ordinates. The elevation of this point (A on the diagram overleaf) above the

chosen datum is equal to the term c in the equation. The slope  

� 

dy
dx

 of any tangent is equal to

2ax + b. But as at Point A, x = 0, the term b in the general equation is the slope or gradient at
point A, the tangent gradient p.

The second derivative 

� 

d 2y
dx 2

 of the general parabola equation equals 2a, a constant. This means

that a tangent to a vertical-axis parabola changes a constant amount of grade for each increment
of distance. (In contrast, a tangent to a vertical circular curve changes direction a constant amount
of angle for equal increments of distance along the arc.) The useful consequence is that the rate of
change of grade on a vertical curve is constant and equals 2a per 100 units of distance (feet or
meters, depending upon the units chosen). On a vertical curve the total change in direction
between the profile grades is  q – p, termed A.

If this change is accomplished on a curve L units long, the constant rate of change must be:

2a  =  

� 

q − p
L

  =  

� 

A
L

with gradients q, p and A in decimal form, and L in either feet or meters, consistent with the job.

A practical formula for a general parabola for a vertical curve is therefore

Elevation  =  a (RD – RDA)2  +  p (RD  –  RDA)  +  ElevationA

where RD is the running distance of any point, RDA is the running distance of point A, a is the
first parameter of the parabola (half the rate of change of grade as a decimal), p is the slope of the
entry (or incoming) tangent (in decimal form), Elevation is the elevation of the point at running
distance RD, and ElevationA is the Elevation of point A. Elevation and RD should be in the same
units, consistent with the entire job.

The location of the turning point of the parabola (the apex) can be computed by noting that the
slope of the tangent at the turning point is 0, and solving 2ax + b = 0. In this case, the location of

the turning point is: x = 

� 

−b
2a

. This value can be converted to a running distance by adding the

value of RDA, and then used in the equation to compute the elevation of the maximum or
minimum point on the curve.

Given the Elevation of a point on the curve, its location, x, from Point A can be computed by
solving the equation:

a x2  +  p x  +  (ElevationA – Elevation)  =  0

As this is a quadratic, the standard quadratic solution will produce two solutions (in most cases),
and it is up to the user to decide which is the most applicable. The program deals with the various
cases (0, 1 and 2 solutions) separately. If there is a single solution, the apex (or nadir) of the curve
has been selected. If there are no solutions, then the elevation selected is beyond the turning point
elevation of the curve and therefore cannot lie on the curve.

For determining the elevations at set distances along the curve, the user can specific an initial
increment, to bring the steps onto an even running distance, then specify a general increment,
which will be used for the remainder of the curve. The program calculates the elevation of the
first point on the curve (the left-hand tangent point), then moves along the initial increment, then
proceeds along the curve using steps of the general increment, until the end of the curve is
reached. The final point on the curve (the right-hand tangent point) is calculated and the program
ends. For each point, the running distance and elevation are shown.

Note that if the final point calculated by increments happens to also be the end point, it will be
calculated and displayed, and then the end point will be calculated again and displayed. This is
because the test for coming to the end of the curve is that the increment is beyond the end point.

Running the Program

Key in XEQ  P  then press the Enter key. The program starts and displays:

COMP  V  CURVE

then prompts for the running distance of the intersection point of the two grade lines, displaying:

ENTER  P  I  RD

then stops while displaying:

R?
0.0000

Key in the RD of the intersection point and press R/S. The calculator then prompts for the
elevation of the intersection point of the two grade lines, displaying:

ENTER  P  I  EL

then stops while displaying:

S?
0.0000

Key in the elevation of the intersection point, then press R/S. The calculator then prompts for the
starting grade (p), displaying:

START  GRADE

then stops while displaying:

P?
0.0000

The value of the slope coming in to the vertical curve should be entered as a decimal value, e.g.,
0.025 (an upward 2.5% gradient), then press R/S. The calculator then prompts for the grade
coming out of the vertical curve, displaying:

END  GRADE

then stops while displaying:

Q?
0.0000

Key in the value of the slope of the grade coming out of the curve, as a decimal grade, e.g., –0.04
(a downward 4% gradient), then press R/S. The calculator then prompts for the length of the
curve, briefly displaying:

ENTER  LENGTH

then stops while displaying:

L?
0.0000

Key in the length of the curve desired and press R/S.

From this point on, the program asks if the user wants to do optional computations. These options
are presented in the order that follows, and any or all of them may be skipped or run.

A. Maximum / Minimum Point

The calculator display  RUNNING  briefly, then prompts for whether or not to calculate the point
on the (whole) parabola where the maximum or minimum elevation occurs. If the vertical curve is
a crest or sag, the maximum or minimum point will occur on the actual curve segment. If the in
and out grades have the same sign, the maximum or minimum point will not be on the actual
curve segment, so this option can be skipped. The calculator briefly displays:

MAX–MIN  [0–1]

then stops and displays the prompt:

Y?
0.0000

The 0 means ‘no’ to the question of whether to run this option, and is the default response. If you
don’t want to run this option, press R/S and the program will advance to the next option. If you

do want to run this option, key in 1 (or any number greater than 0), which signifies ‘yes,’ and
press R/S. The calculator briefly displays:

MAX–MIN  EL

then stops and displays the Elevation of the maximum or minimum point, looking as follows:

E=
      114.5611

Press R/S to continue. The calculator then briefly displays:

MAX–MIN  RD

then stops and displays the Running Distance of the maximum or minimum point, looking as
follows:

D=
      24,589.2222

Press R/S to continue, and this option concludes.

B. Compute the End Points

This option computes the running distance and elevation of the end points of the curve, where the
parabolic curve joins the tangent gradients. The calculator briefly displays:

END  PTS  [0–1]

then stops and displays the prompt:

Y?
0.0000

If you don’t want to run this option, press R/S and the program will advance to the next option. If
you do want to run this option, key in 1 (or any number greater than 0), which signifies ‘yes,’ and
press R/S. The calculator briefly displays:

START  PT  RD

then stops and display the running distance of the start point of the curve, such as follows:

U=
       24,367.0000

Press R/S to continue. The calculator briefly displays:

START  PT  EL

then stops and displays the elevation of the start point of the curve, such as follows:

V=

         103.4500

Press R/S to continue. The calculator briefly displays:

END  PT  RD

then stops and display the running distance of the end point of the curve, such as follows:

F=
       24,767.0000

Press R/S to continue. The calculator briefly displays:

END  PT  EL

then stops and displays the elevation of the end point of the curve, such as follows:

E=
            107.4500

Press R/S to continue, and the option ends.

C. Compute Elevation at a Specified Running Distance

With this option, the user can enter any running distance and the calculator will compute the
elevation on the curve at that point. The calculator does not check if the running distance is on the
curve segment actually being used, so the user must check this. The calculator briefly displays:

COMP  1  EL  [0–1]

although the right-hand bracket will be off screen. The calculator then stops and displays:

Y?
0.0000

If you don’t want to run this option, press R/S and the program will advance to the next option. If
you do want to run this option, key in 1 (or any number greater than 0), which signifies ‘yes,’ and
press R/S. The calculator briefly displays:

ENTER  RD

then stops and prompts for the RD to be entered, displaying:

X?
0.0000

Key in the running distance of the point of interest and press R/S. The calculator then briefly
displays:

EL  OF  POINT

then stops and displays the elevation value of the selected point, such as follows:

H=
           114.3690

Press R/S to continue, and the calculator prompts to see if you want to compute another elevation,
briefly displaying:

AGAIN  [0–1]

then stopping and displaying:

Y?
0.0000

If you want to do another point, key in 1 and press R/S. The program then prompts for the
running distance (as above) and loops through the option until you decide not to do it again. If
you don’t want to do this option, press R/S, and the option ends.

D. Compute the Running Distances at a Specified Elevation

This option allows the user to enter an elevation and compute the running distance(s) at which it
occurs. Since the curve is a parabola, there will be zero, one or two solutions. If the elevation
cannot occur on the curve, i.e., the elevation is beyond the elevation of the turning point of the
parabola, there will be zero answers.

If the elevation chosen is that of the turning point, there will be just one answer, which will be
provided. Any other elevation on the curve will have two solutions, and each will be given.

The calculator briefly displays:

COMP  1  RD  [0–1]

with the right-hand bracket off the screen. The calculator then stops and displays the prompt:

Y?
0.0000

If you want to skip this option, just press R/S and the calculator moves on to the next option. If
you do want to run this option, key in 1 and press R/S. The calculator briefly displays:

ENTER  EL

then stops and displays:

H?
0.0000

Key in the elevation of interest, and press R/S.

If the elevation is not on the curve, the calculator briefly displays:

THIS  ELEV

then

NOT  ON  CURVE

and then prompts to see if you want to run the option again.

If the elevation is at the maximum or minimum point, the calculator briefly displays:

SINGLE  RD

then stops and displays the running distance of the point, such as follows:

X=
      24,392.9180

Press R/S/ to continue and the calculator prompts to see if you want to run the option again.

If the elevation is elsewhere on the curve, the calculator briefly displays:

FIRST  RD

then stops and displays the running distance of the first solution point, such as follows:

X=
      24,383.0819

Press R/S and the calculator briefly displays:

SECOND  RD

then stops and displays the running distance of the second solution point, such as follows:

X=
       24,795.3625

It is up to the user to decide if the points fall within the end points of the curve, and chose points
that are useful.

Press R/S to continue, and the calculator then briefly displays:

AGAIN  [0–1]

then stops and displays:

Y?
0.0000

To run the option again, key in and 1 and press R/S. The calculator will then prompt for the
elevation to be entered, as above, and run through the option again. If you don’t want to run the
option again, press R/S and the option ends.

E. Step Through a Series of Running Distances to get Elevations at each

The final option allows the user to step through a series of equally-spaced points along the curve.
As the start point is often at an odd running distance, this option allows the user to select a first
increment, to allow the running distances to be brought to even values (e.g., exactly onto 100 ft
stations), and then select a general increment to be applied successively until the end point is
reached. The end point is calculated as the last point along the curve.

The option begins by displaying briefly:

STEP  THRU  RD

then displaying briefly:

NO–YES  [0–1]

then stopping and displaying:

Y?
0.0000

If you want to run this option, key in 1 and press R/S. If not, just press R/S and the program ends.
If you are running the option, the calculator briefly displays:

FIRST  INCRMNT

then stops and displays:

C=
0.0000

Key in the first increment, then press R/S. The calculator briefly displays:

GENRL  INCRMNT

then stops and displays:

D=
0.0000

Key in the increment to be used for all the other distances, then press R/S. The calculator then
briefly displays:

RD  VALUE

then stops and displays the running distance of the start point, such as follows:

U=
      24,367.0000

Press R/S to continue. The calculator briefly displays:



HP-35s Calculator Program Curves 2A
Compute Values for a Parabolic Vertical Curve

- 16 -

EL  VALUE

then stops and displays the elevation at the start point, such as follows:

V=
        103.4500

Press R/S to continue. The calculator then loops through the following sequence, briefly
displaying:

RD  VALUE

then stopping and displaying the next running distance value, such as follows:

X=
     24,467.0000

Press R/S to continue. The calculator briefly displays:

EL VALUE

then stops and displays the elevation value at the running distance just given, such as follows:

H=
          111.2000

Press R/S to continue through this loop until the increments extend past the last point. At this
stage, the last point is displayed. The calculator briefly displays:

END  POINT

then briefly displays:

RD VALUE

then stops and displays the running distance of the end point, such as follows:

X=
       24,767.0000

Press R/S to continue. The calculator briefly displays:

EL  VALUE

then stops and displays the elevation of the end point, such as follows:

H=
         107.45

Press R/S to continue. The calculator briefly displays

PROGRAM  END

and then stops at the end of the program, having reset Flag 10 to its value at the time the program
was started.. If R/S is pressed again, the program returns to the point whence it was called. This is
not strictly necessary, unless the program was called by another program.

Sample Computations

The sample computations are based on the following general data:

Running Distance of the Point of Intersection  =  24,567.0000

Elevation of the Point of Intersection  =  123.4500

Length of the Curve  =  400.0000

The running distance of the start point is therefore 24,367 and of the end point is 24,767.

The following tabulations show the results for different in and out gradients. The increments used
for the running distance values are both 50 (both the first increment and the general increment).

Case In Grade Out Grade Start EL End EL Max/Min RD Max/Min EL

1 0.0200 -0.0200 119.45 119.45 24567.00 121.45
2 0.0400 -0.0600 115.45 111.45 24527.00 118.65
3 0.0600 -0.0400 111.45 115.45 24607.00 118.65
4 0.0800 -0.1000 107.45 103.45 24544.78 114.56
5 0.1000 -0.0800 103.45 107.45 24589.22 114.56
6 -0.0200 0.0400 127.45 131.45 24500.33 126.12
7 -0.0400 0.0200 131.45 127.45 24633.67 126.12
8 -0.0600 0.0800 135.45 139.45 24538.43 130.31
9 -0.0800 0.1000 139.45 143.45 24544.78 132.34

10 -0.1000 0.0600 143.45 135.45 24617.00 130.95
11 0.0200 0.0400 119.45 131.45 23967.00 115.45
12 0.0400 0.0200 115.45 127.45 25167.00 131.45
13 0.0600 0.0800 111.45 139.45 23167.00 75.45
14 0.0800 0.1000 107.45 143.45 22767.00 43.45
15 0.1000 0.0600 103.45 135.45 25367.00 153.45
16 -0.0200 -0.0400 127.45 115.45 23967.00 131.45
17 -0.0400 -0.0600 131.45 111.45 23567.00 147.45
18 -0.0600 -0.0400 135.45 115.45 25567.00 99.45
19 -0.0800 -0.1000 139.45 103.45 22767.00 203.45
20 -0.1000 -0.0800 143.45 107.45 26367.00 43.45

Running Distance
Case 24367 24417 24467 24517 24567 24617 24667 24717 24767

1 119.45 120.33 120.95 121.33 121.45 121.33 120.95 120.33 119.45
2 115.45 117.14 118.20 118.64 118.45 117.64 116.20 114.14 111.45
3 111.45 114.14 116.20 117.64 118.45 118.64 118.20 117.14 115.45
4 107.45 110.89 113.20 114.39 114.45 113.39 111.20 107.89 103.45
5 103.45 107.89 111.20 113.39 114.45 114.39 113.20 110.89 107.45
6 127.45 126.64 126.20 126.14 126.45 127.14 128.20 129.64 131.45
7 131.45 129.64 128.20 127.14 126.45 126.14 126.20 126.64 127.45
8 135.45 132.89 131.20 130.39 130.45 131.39 133.20 135.89 139.45
9 139.45 136.01 133.70 132.51 132.45 133.51 135.70 139.01 143.45

10 143.45 138.95 135.45 132.95 131.45 130.95 131.45 132.95 135.45
11 119.45 120.51 121.70 123.01 124.45 126.01 127.70 129.51 131.45
12 115.45 117.39 119.20 120.89 122.45 123.89 125.20 126.39 127.45
13 111.45 114.51 117.70 121.01 124.45 128.01 131.70 135.51 139.45
14 107.45 111.51 115.70 120.01 124.45 129.01 133.70 138.51 143.45
15 103.45 108.33 112.95 117.33 121.45 125.33 128.95 132.33 135.45
16 127.45 126.39 125.20 123.89 122.45 120.89 119.20 117.39 115.45
17 131.45 129.39 127.20 124.89 122.45 119.89 117.20 114.39 111.45
18 135.45 132.51 129.70 127.01 124.45 122.01 119.70 117.51 115.45
19 139.45 135.39 131.20 126.89 122.45 117.89 113.20 108.39 103.45
20 143.45 138.51 133.70 129.01 124.45 120.01 115.70 111.51 107.45

These values should allow the program to be tested to make sure it is working properly. The
above tabulated values were calculated by spreadsheet, rather than the calculator, but the
calculator results were checked against the these tabulations.

Storage Registers Used

A Difference between the incoming and outgoing gradients.

B Parameter a in the parabola equation.

C Elevation difference. First increment value for ‘stepping’ option.

D Running Distance of a computed point. General increment for ‘stepping’ option.

E Elevation of a computed point.

F Running Distance of a computed point.

H Elevation of a computed point, and entered elevation to have RD calculated.

I Distance along curve from start point.

L Length of the curve to be computed.

P Gradient of the incoming tangent (start grade)

Q Gradient of the outgoing tangent (end grade).

R Running Distance of the Point of Intersection.

S Elevation of the Point of Intersection.

U Running Distance of the start point.

V Elevation of the start point.

X Running Distance entered to compute elevation at that point.

Y Yes/No variable for option choices.

Statistical Registers: Not used.

Labels Used

Label P Length  =  1383 Checksum  =  6E0F

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computation to check proper operation after entry.

Flags Used

Flags 1 and 10 are used by this program. Flag 10 is set for this program, so that equations can be
shown as prompts. Flag 1 is used to record the setting of Flag 10 before the program begins. At
the end of the program, Flag 10 is reset to its original value, based on the value in Flag 1.


